Too Few Parameters

by Roger Carlson

There are a variety of circumstances under which you might want to open a query in code.  You may need to open a recordset to do some sort of complex processing that can't be done directly in SQL.  Or you might need to run an Action Query from a button. Microsoft Access provides a way to do these things through its Data Access Object model or DAO.
Generally, it is as simple as defining object variables for the database and recordset, setting the database variable to the current database.
Dim db As DAO.Database

Dim rs As DAO.Recordset
    Set db = CurrentDb()
Then using the OpenRecordset method of the database object to open the query.
    Set rs = db.OpenRecordset("Query1")
Or  using the Execute method of the database object to run the action query

    db.Execute "Query2", dbFailOnError 

This works fine unless your query is a Parameter Query.  Then you will get the "Too few parameters. Expected xx" error.  

A parameter query is method provided by SQL to allow the user to interact with a query.  It is accomplished by adding a prompt in brackets in the Where Clause of the SQL statement or the Criteria row of the query in the Query Builder.  For instance, if you had a table called Books and you wanted to only display those books by a certain publisher.  You could create a query that looked something like this:
SELECT ISBN, Title, PubID, Price

FROM Books

WHERE PubID=[Enter publisher];
When you run this query, you will get a dialog box like Figure 1 that will allow you to enter the publisher you want.  The value in the brackets becomes the prompt message in the dialog box.

Dialog box that allows the user to supply a value for the publisher.

As useful as this is, it can be clumsy when there are several parameters in the query.  In that case, it is usually best to create a form with textboxes that hold the values for the query and then run the query from that form.  The query will then reference the controls on that form as its parameter values.
For instance, suppose we had a form (called Form1 here for convenience) with two text boxes (Text1 and Text2) so we could select books by either Title or Publisher.  See Figure 2.

Form used as a parameter dialog for complex parameter.
I could then modify the query above to reference the two text boxes on the form, like this:

SELECT ISBN, Title, PubID, Price

FROM Books

WHERE Title=[forms]![form1]![text1] OR PubID=[forms]![form1]![text2];
The confusing part is this:  if I only wanted to run and display the query, I could use a command like the following in the code behind the button.

DoCmd.OpenQuery "Query1"
And the query would display just as if I'd run it from the query window.  But if I tried to open a recordset based on the same query, like this:

    Set rs = db.OpenRecordset("Query1")
I would get the Two Few Parameters error,

Why?  Well, it is important to remember that Access is not a single program, but has separate pieces.   There is Access the Graphical User Interface (GUI) that allows you to create tables, queries, form, reports and such.  Then there is the Jet database engine, which is the SQL interpreter.  And lastly, there is Visual Basic for Applications (VBA) which can include a variety of components, DAO being one of them.

The Access GUI and SQL work pretty well together.  That's why the Docmd.OpenQuery statement works.  As long as the form is open, SQL will use the text boxes on the form as the parameters for the query.

However, VBA (and particularly DAO) needs some additional information in order to read the text boxes on the form.  And for this, we need to talk about QueryDefs.
The QueryDef object of the DAO object model allows you to create, manipulate, and reference saved queries in code.  This includes the ability to read the parameter from the GUI.  

To use the QueryDef object, you must declare and assign an object variable, just as you do for the database object and the recordset object.  
Dim qdf As DAO.QueryDef
Set qdf = db.QueryDefs("Query1")
Now we're already to use the QueryDef to read the parameters.  Access actually provide two ways to do this.  You can either directly reference the parameters or you can use the Parameters collections (another object of the DAO object model).  So let's look at both.

Direct Reference

As the name implies, the direct reference method explicitly references the parameters in the query.  To directly reference the parameter, you follow the object variable with a "bang" (!) and then the actual parameter from the query.  Then you assign it the value from the text box on your form.  Like this:

    qdf![Forms!form1!Text1] = [Forms]![form1]![Text1]
Putting it all together, it looks like this:

Dim db As DAO.Database

Dim rs As DAO.Recordset

Dim qdf As DAO.QueryDef

    Set db = CurrentDb()

    Set qdf = db.QueryDefs("Query1")

    qdf![Forms!form1!Text1] = [Forms]![form1]![Text1]
    qdf![Forms!form1!Text2] = [Forms]![form1]![Text2]

   Set rs = qdf.OpenRecordset(dbOpenDynaset)
The following syntaxes also work for referencing the parameters:

    qdf("[Forms]![form1]![Text1]") = [Forms]![form1]![Text1]

    qdf("[Forms]![form1]![Text2]") = [Forms]![form1]![Text2]

or
    qdf(0) = [Forms]![form1]![Text1]

    qdf(1) = [Forms]![form1]![Text2]

Parameters Collection
The QueryDef object also has a Parameters Collection that will hold the values of the parameters.  

In order to use the parameters collection, you need to declare yet another variable.
    Dim prm As Parameter
Then you use it, you create a loop For…Next loop to read through all the parameters in the collection.  This loop uses the Eval function to read the parameter and assign it to the parameter variable.

    For Each prm In qdf.Parameters

        prm.Value = Eval(prm.Name)

    Next prm
Putting this one together, we get this:

    Dim db As DAO.Database

    Dim rst As DAO.Recordset

    Dim qdf As DAO.QueryDef

    Dim prm As Parameter

    Set db = CurrentDb

    Set qdf = db.QueryDefs("Query1")

    For Each prm In qdf.Parameters

        prm.Value = Eval(prm.Name)

    Next prm

    Set rst = qdf.OpenRecordset(dbOpenDynaset)

The advantage of this method is that it is not tied to a specific query.  The same code can be used to read the parameters of any parameter query, no matter how many there are or what they're called.  You could even use the same subroutine (with some modifications) to open different parameter queries.

